
Subject: Re: 2D to 3D Array
Posted by David Fanning on Fri, 07 Apr 2006 20:19:47 GMT
View Forum Message <> Reply to Message

rafaloos@gmail.com writes:

>  I have a Band Interleaved Image (BIL):
>  samples = 300
>  lines = 316
>  bands = 481
>  
>  I can read this image as a 2D array of (144300, 316) Where 144300 will
>  be the product of samples by bands.
>  I would like to reformat the image so that I can put the data into an
>  array such as (300, 316, 481).
>  
>  For example: I have this array (6,2)
>  
>  array[*,0] = [1, 2, 1, 2, 1, 2]
>  array[*,1] = [3, 4, 3, 4, 3, 4]
>  
>  What I want is a result as:
>  
>  array[*,*,0] = 1, 2
>                 3, 4
>  
>  array[*,*,1] = 1, 2
>                 3 ,4
>  
>  array[*,*,2] = 1, 2
>                 3, 4
>  
>  so that my array will be (2,2,3) 

First reform your array into a [2,3,2] array:

   array = Reform(array, 2, 3, 2)

Then, transpose the dimensions the way you want them:

   array = Transpose(array, [0,2,1])

IDL> print, array[*,*,2]
       1       2
       3       4
IDL> print, array[*,*,1]
       1       2
       3       4

IDL> print, array[*,*,0]
       1       2
       3       4

Cheers,

David

-- 
David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: http://www.dfanning.com/

Subject: Re: 2D to 3D Array
Posted by Paul Van Delst[1] on Fri, 07 Apr 2006 20:21:22 GMT
View Forum Message <> Reply to Message

rafaloos@gmail.com wrote:
>  I have a Band Interleaved Image (BIL):
>  samples = 300
>  lines = 316
>  bands = 481
>  
>  I can read this image as a 2D array of (144300, 316) Where 144300 will
>  be the product of samples by bands.
>  I would like to reformat the image so that I can put the data into an
>  array such as (300, 316, 481).
>  
>  For example: I have this array (6,2)
>  
>  array[*,0] = [1, 2, 1, 2, 1, 2]
>  array[*,1] = [3, 4, 3, 4, 3, 4]
>  
>  What I want is a result as:
>  
>  array[*,*,0] = 1, 2
>                 3, 4
>  
>  array[*,*,1] = 1, 2
>                 3 ,4
>  
>  array[*,*,2] = 1, 2
>                 3, 4
>  
>  so that my array will be (2,2,3) 

REFORM your [144300,316] array to [300,481,316], and then use TRANSPOSE to swap the last 

two indices. E.g.

IDL> array=lonarr(8,2)
IDL> array[*,0] = [1, 2, 1, 2, 1, 2, 1, 2]
IDL> array[*,1] = [3, 4, 3, 4, 3, 4, 3, 4]
IDL> help, array
ARRAY           LONG      = Array[8, 2]
IDL> a=reform(array,2,4,2)
IDL> help, a
A               LONG      = Array[2, 4, 2]
IDL> print, a[*,*,0]
            1           2
            1           2
            1           2
            1           2
IDL> print, a[*,*,1]
            3           4
            3           4
            3           4
            3           4
IDL> b=transpose(a,[0,2,1])
IDL> help, b
B               LONG      = Array[2, 2, 4]
IDL> print, b[*,*,0]
            1           2
            3           4
IDL> print, b[*,*,1]
            1           2
            3           4
IDL> print, b[*,*,2]
            1           2
            3           4
IDL> print, b[*,*,3]
            1           2
            3           4

paulv

-- 
Paul van Delst
CIMSS @ NOAA/NCEP/EMC

Subject: Re: 2D to 3D Array

Posted by rafaloos on Fri, 07 Apr 2006 20:41:05 GMT
View Forum Message <> Reply to Message

Thanks ...
It perfectly worked ... (:

Subject: Re: 2D to 3D Array
Posted by David Fanning on Fri, 07 Apr 2006 20:57:56 GMT
View Forum Message <> Reply to Message

rafaloos@gmail.com writes:

>  Thanks ...
>  It perfectly worked ... (:

It pays to read the Dimensional Juggling Tutorial once
in awhile. :-)

Cheers,

David
-- 
David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: http://www.dfanning.com/

